SESUG 2022 Paper 178

A SAS Macro to Calculate Blinding Index in Clinical Trials: %blinding_index, an application of PROC IML

Jingwei Wu, Di Zhu, Temple University

ABSTRACT

Randomized clinical trials are often considered as gold standard for clinical research, owing the fact of its rigorous study design and implementation. Success of blinding, i.e., study participants and key layperson are unaware of the treatment assignment or therapeutic allocation they received, is key component to minimize post-randomization bias. There are two commonly used methods that can quantify blinding success in a double-blinded randomized control trial based on administered post-randomization questionnaire data, namely, James Blinding Index (James_BI) and Bang Blinding Index (Bang_BI). James_BI is a scaled number between 0 and 1, increases as the success of blinding increases. Bang_BI is calculated per treatment arm, is a scaled number between -1 to 1, with 0 as the most desirable situation under successful blinding.

There is no current SAS macro program that can calculate the two blinding indexes. To fit this gap, we have written a SAS macro: %blinding_index to provide researchers a computational tool to calculate these blinding indexes. We used PROC IML to calculate the indexes, along with the corresponding 95% confidence intervals for statistical inferences. Researchers may thus use it to assess successfulness of blinding in clinical trials. This presentation will review the two methods for the assessment of blinding and demonstrate the developed macro with two trial applications.

INTRODUCTION

Randomized clinical trials are often considered as gold standard for clinical research, owing the fact of its rigorous study design and implementation. Even though randomization reduces the chance of selection bias and minimizes the influence of confounding variables at the onset of the trial, it will not preclude biased outcome assessment at follow-up visits (Karanicolas et al.) . Success of blinding, i.e., study participants and key layperson are unaware of the treatment assignment or therapeutic allocation, is key component to minimize post-randomization bias. Otherwise, study participants may inevitably change their behaviors if they know what treatment they receive, and key layperson (for example, treating clinicians or outcome assessors) may intentionally ascertain outcome evaluations with biased opinions if they know from which treatment groups these outcomes collect. The inherent risk in "unsuccessful blinding" is undeniable since it may easily induce more ascertainment bias and worse compliance.

While various strategies on planning the use of blinding in randomized clinical trials are widely disseminated in the past few decades, it is also important to strive to evaluate the successfulness of blinding at the end of study. There are two commonly used methods that can quantify blinding success in a double-blinded randomized control trial based on administered post-randomization questionnaire data, namely, James Blinding Index (BI_James) and Bang Blinding Index (BI_Bang). BI_James is a scaled number between 0 and 1, increases as the success of blinding increases. BI_Bang is calculated per treatment arm, is a scaled number between -1 to 1, with 0 as the most desirable situation under successful blinding.

There is no current SAS macro program that can calculate the two blinding indexes. To fit this gap, we have written a SAS macro: `%blinding_index` to provide researchers a computational tool to calculate these blinding indexes. We used PROC IML to calculate the indexes, along with the corresponding 95% confidence intervals for statistical inferences. Researchers may thus use it to assess successfulness of blinding in clinical trials.

METHODS

BLINDING ASSESSMENT QUESTIONNAIRE

Commonly, study participants (or laypersons) are asked to fill in a blinding questionnaire at the end of the study. In the questionnaire, participants are asked to guess which treatment assignment they received. They can choose from three responses: "Drug", "Placebo", or "Don't Know". A typical format of the response data is presented in Table 1.

Assignment	**Response**		
	Drug	Placebo	Don't know
Drug	n_{11}	n_{12}	n_{13}
Placebo	n_{21}	n_{22}	n_{23}

Table 1. Cross-tabulate format to summarize responses by treatment assignment and guess

JAMES BLINDING INDEX (BI_JAMES)

James et al. proposed the construction of an index of blindness, which takes into account the "Don't Know" response that is most indicative of blinding and places more weight on this desirable response (James et al.). It is, in fact, a variation of a Kappa coefficient that is sensitive not to the degree of agreement, but to the degree of disagreement. This index ranges from 0 to 1: 0 indicates a complete unblinding (i.e., all participants guessed their treatment assignments correctly), 1 indicates a complete blinding (i.e., all participants guessed their treatment assignments incorrectly), 0.5 indicates a random blinding (i.e., half of the participants guessed their treatment assignments correctly, while the other half guessed incorrectly). If the upper limit of the two-sided confidence interval of BI_James is < 0.5, then the study will be claimed unblinding.

BANG BLINDING INDEX (BI_BANG)

Bang et al. proposed a new way to assess success of blinding, which not only takes into account the influence of uncertain responses, but also be able to distinguish the different blinding performances in each study arm (Bang et al.). It calculates separate blinding index per treatment arm, thus, has the ability to provide the proportion of the unblinding in each arm. It takes a value between -1 to 1: a positive value indicates a possible unblinding beyond chance (i.e., a majority of participants guessed their treatment assignment correctly), a negative value indicates possibly either a success of blinding or a failure of blinding in the opposite direction (i.e., a majority participants incorrectly named alternative treatment assignments), 0 indicates a random blinding (i.e., half of the participants guessed their treatment assignments correctly, while the other half guessed incorrectly). If the lower limit of the one-sided confidence interval of BI_Bang is > 0, then the study will be claimed unblinding.

SAS MACRO IMPLEMENTATION

MACRO REQUIREMENT

The SAS macro, `%blinding_index`, requires two parameters. The first parameter, `count`, contains a list of the number of participants who choose responses from "Drug", "Placebo", or "Don't Know", separated by the actual treatment assignments they received (i.e., enter columnwise cell counts, separated by comma). The second parameter, `arm`, is used to define the names of the actual treatment assignments.

The following macro call will calculate the two blinding indexes based on Table 1 tabulation.

```
%blinding_index(count=%str({n11 n21, n12 n22, n13 n23}),
                arm=%str({Drug, Placebo}));
```

MACRO PROGRAM

The macro program is based on PROC IML procedure. It includes different steps to compute two blinding indexes and produce outputs.

1. Read data into PROC IML

```
%macro blinding_index(count=, arm=);
proc iml;
  counts = &count.;
  arms = &arm.;
```

2. Calculate James Blinding Index and its two-sided confidence intervals

```
/* James BI Calculation */
   weights = {0 0.5,
              0.5 0,
              1 1};
   p = counts / sum(counts);
   rowT_James = p[,+];
   p1 = p || rowT_James;
   colT_James = p1[+,];
   p2 = p1 // colT_James;

   pdk = p2[nrow(p2)-1, ncol(p2)];
   pdo = 0;
   tmp1 = 0;
   do i =1 to nrow(p2)-2;
      do j=1 to ncol(p2)-1;
           pdo = pdo+weights[i,j]*p2[i,j]/(1-pdk);
             tmp1 = tmp1+weights[i,j]*p2[i,ncol(p2)]*(p2[nrow(p2),j]-
                    p2[nrow(p2)-1,j]);
      end;
   end;
   pde = tmp1/(1-pdk)##2;
   kd = (pdo-pde)/pde;
   BI_James = (1+pdk+(1-pdk)*kd)/2;

   denom1 = 4*tmp1##2;
   num1 = 0;
   do i=1 to nrow(p2)-2;
      do j=1 to ncol(p2)-1;
         tmp2 = 0;
             do r=1 to ncol(p2)-1;
```

```
                tmp2 = tmp2+(weights[r,j]*p2[r,ncol(p2)]+weights[i,r]*
                       (p2[nrow(p2),r]-p2[nrow(p2)-1,r]));
             end;
          num1 = num1+p2[i,j]*(1-pdk)##2*(weights[i,j]*(1-pdk)-
                 (1+kd)*tmp2)##2;
      end;
   end;

   v1 = num1/denom1;
   v2 = pdk*(1-pdk)-(1-pdk)*(1+kd)*(pdk+(1-pdk)*(1+kd)/4);

   BI_var = (v1+v2)/sum(counts);
   BI_James_se = sqrt(BI_var);
   BI_James_Lower = BI_James-1.96*BI_James_se;
   BI_James_Upper = BI_James+1.96*BI_James_se;
```

3. Calculate Bang Blinding Index and its one-sided confidence intervals

```
/* Bang BI Calculation */
   t_counts = counts`;
   rowT_Bang = t_counts[,+];
   t_counts1 = t_counts || rowT_Bang;
   colT_Bang = t_counts1[+,];
   t_counts2 = t_counts1 // colT_Bang;

   BI_Bang = {0 0};
   BI_Bang_se = {0 0};
   do i=1 to nrow(t_counts2)-1;
      BI_Bang[i] = (2*t_counts2[i,i]/(t_counts2[i,1]+t_counts2[i,2])-1)
                   *(t_counts2[i,1]+t_counts2[i,2])
                   /t_counts2[i,ncol(t_counts2)];
      BI_Bang_se[i] = sqrt((t_counts2[i,1]/t_counts2[i,ncol(t_counts2)]*(1-
                      t_counts2[i,1]/t_counts2[i,ncol(t_counts2)])
                     +t_counts2[i,2]/t_counts2[i,ncol(t_counts2)]*(1-
                      t_counts2[i,2]/t_counts2[i,ncol(t_counts2)])
                      +2*t_counts2[i,1]/t_counts2[i,ncol(t_counts2)]
                      *t_counts2[i,2]/t_counts2[i,ncol(t_counts2)])
                     /t_counts2[i,ncol(t_counts2)]);
   end;

   BI_Lower = BI_Bang-1.645*BI_Bang_se;
   BI_Upper = BI_Bang+1.645*BI_Bang_se;
   BI_Bang = BI_Bang`;
   BI_Bang_Lower = BI_Lower`;
   BI_Bang_Upper = BI_Upper`;
```

4. Produce summary outputs

```
/* Summary Output */
   title "Blinding Index Summary";
   if BI_James ^=. then
      print BI_James BI_James_Lower BI_James_Upper;
   if BI_Bang ^=. then
      print Arms BI_Bang BI_Bang_Lower BI_Bang_Upper;
   run;
   quit;
   %mend;
```

PRACTICAL EXAMPLES

EXAMPLE 1

The Cholesterol Reduction in Seniors Program (CRISP) pilot study was a randomized, double-blinded clinical trial to assess feasibility of recruitment and efficacy of cholesterol lowering in people over 65 years old. 431 subjects with low-density lipoprotein cholesterol levels greater than 4.1 and less than 5.7 mmol/L were randomized into placebo, 20-mg Lovastatin and 40-mg Lovastatin arms. Detailed trial information can be found in the published article (LaRosa et al.).

At the end of the trial, all subjects were asked to rate whether they knew their medications on a five-point scale: (1) strongly believe the treatment is Lovastatin; (2) somewhat believe the treatment is Lovastatin; (3) somewhat believe the treatment is placebo; (4) strongly believe the treatment is placebo; (5) Don't Know. 416 subjects returned the questionnaire, thus, was considered in the blinding assessment analysis. To simplify further, subjects' responses in "strongly believe" or "somewhat believe" were merged, and two doses were combined as a single "Lovastatin" arm. Table 2 presents the CRISP study data {Bang et al.}.

Assignment	**Response**		
	Lovastatin	Placebo	Don't know
Lovastatin	82	25	170
Placebo	27	29	83

Table 2. CRISP Study Data

We call our macro to calculate two the two blinding indexes.

```
%blinding_index (count=%str({82 27, 25 29, 170 83}),
                 arm=%str({Lovastatin,Placebo}));
```

An output is generated:

Blinding Index Summary

BI_James	BI_James_Lower	BI_James_Upper
0.7479275	0.7053222	0.7905328

Arms	BI_Bang	BI_Bang_Lower	BI_Bang_Upper
LOVASTATIN	0.2057762	0.1478112	0.2637412
PLACEBO	0.0143885	-0.07415	0.1029273

Output 1. Output of CRISP Study

James's blinding index equals to 0.75 (95% CI: 0.71, 0.78). Since the two-sided confidence interval does not include 0.5, we may conclude that the CRISP study was well-blinded. Bang's blinding index equals to 0.21 (95% CI: 0.15, 0.26) for the Lovastatin arm, and 0.01 (95% CI: -0.07, 0.10) for the placebo arm, respectively. It implies that approximately 21% of subjects correctly guessed their treatments beyond random chance in the Lovastatin arm, whereas only roughly 1% of subjects did in the placebo arm. Furthermore, by comparing lower limit of confidence interval with 0, we may conclude that subjects in the Lovastatin

arm were not blinded well, while subjects in the placebo arm were well-blinded. These results are consistent with the findings in the published article (Bang et al.).

EXAMPLE 2

The Pilot Water Evaluation Trial (Pilot WET) was a randomized, triple-blinded home drinking water intervention trial to determine if a large study could be undertaken while successfully blinding participants. 77 Households in northern California were randomized to use externally identical active or sham treatment devices. Detailed trial information can be found in the published article (Colford et al.).

At the end of the trial, one member of each household, designated the "index respondent", was asked to report on a questionnaire one of five possible responses: (1) It is definitely the active water treatment device; (2) It is probably the active water treatment device; (3) It is probably not the active water treatment device; (4) It is definitely not the active water treatment device; (5) I'm not sure. A total of 64 index respondents returned the questionnaire, thus, was considered in the blinding assessment analysis. To accommodate the blinding index, these responses were collapsed to three categories: "The active device," "Not the active device," or "I don't know." Table 3 presents the Pilot WET trial data.

Assignment	**Response**		
	Active	Sham	Don't know
Active	19	5	9
Sham	13	5	12

Table 3. Pilot WET Trial Data

We call our macro to calculate two the two blinding indexes.

```
%blinding_index (count=%str({19 13, 5 5, 9 13}),
                 arm=%str({Active,Sham}));
```

An output is generated:

Blinding Index Summary

BI_James	BI_James_Lower	BI_James_Upper
0.6477482	0.5370941	0.7584022

Arms	BI_Bang	BI_Bang_Lower	BI_Bang_Upper
ACTIVE	0.4242424	0.2123975	0.6360874
SHAM	-0.258065	-0.469894	-0.046235

Output 2. Output from Pilot WET Trial Study

James's blinding index equals to 0.65 (95% CI: 0.54, 0.76). Since the two-sided confidence interval does not include 0.5, we may conclude that the Pilot WET trial was well-blinded. Bang's blinding index equals to 0.43 (95% CI: 0.21, 0.64) for the active device group, and -0.25 (95% CI: -0.47, -0.05) for the sham device group, respectively. It implies that approximately 43% of index respondents correctly guessed their treatment devices beyond random chance in the active device group, whereas 25% of index respondent mistakenly named the alternative treatment devices beyond random chance in the sham device group.

Furthermore, by comparing lower limit of confidence interval with 0, we may conclude that households received active devices were not blinded well, while household received sham devices has a tendency for opposite guesstimate (e.g., 'wishful thinking'). These results are consistent with the findings in the published article (Colford et al.).

CONCLUSION

In randomized clinical trials, blinding is as important as randomization. Quantitative assessment of blinding is necessitated to ensure systemic bias is not migrated to final study findings. This macro program calculates two commonly used blinding indexes, along with the corresponding 95% confidence intervals for statistical inferences. Researchers may use it to assess successfulness of blinding in clinical trials.

REFERENCES

Bang, H., Ni, L., & Davis, C. E. (2004). Assessment of blinding in clinical trials. Controlled Clinical Trials, 25(2), 143–156. https://doi.org/10.1016/j.cct.2003.10.016

Colford, J. M., Rees, J. R., Wade, T. J., Khalakdina, A., Hilton, J. F., Ergas, I. J., Burns, S., Benker, A., Ma, C., Bowen, C., Mills, D. C., Vugia, D. J., Juranek, D. D., & Levy, D. A. (2002). Participant Blinding and Gastrointestinal Illness in a Randomized, Controlled Trial of an In-Home Drinking Water Intervention. Emerging Infectious Diseases, 8(1), 29–36. https://doi.org/10.3201/eid0801.001481

James, K. E., Bloch, D. A., Lee, K. K., Kraemer, H. C., & Fuller, R. K. (1996). An Index for Assessing Blindness in a Multi-Centre Clinical Trial: Disulfiram for Alcohol Cessation—a Va Cooperative Study. Statistics in Medicine, 15(13), 1421–1434. doi: 10.1002/(SICI)1097-0258(19960715)15:13<1421::AID-SIM266>3.0.CO;2-H

Karanicolas, P. J., Farrokhyar, F., & Bhandari, M. (2010). Practical tips for surgical research: blinding: who, what, when, why, how?. Canadian journal of surgery, 53(5), 345–348.

LaRosa, J. C., Applegate, W., Crouse, J. R., 3rd, Hunninghake, D. B., Grimm, R., Knopp, R., Eckfeldt, J. H., Davis, C. E., & Gordon, D. J. (1994). Cholesterol lowering in the elderly. Results of the Cholesterol Reduction in Seniors Program (CRISP) pilot study. Archives of internal medicine, 154(5), 529–539. https://doi.org/10.1001/archinte.154.5.529

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Jingwei Wu
jingwei.wu@temple.edu